HP 48 / HP 50g Calculator Programs

Dr. D.G. Simpson
Department of Physical Sciences and Engineering
Prince George's Community College

May 3, 2014

These calculator programs are written for the Hewlett-Packard HP 50g scientific calculator and other HP calculators that use HP User RPL (e.g. the HP 48 series).

Contents

1 Projectile Problem

This program solves the projectile problem: given a target sitting on a hill at coordinates (x_t, y_t) and a cannon with muzzle velocity v_0, at what angle θ should the cannon be aimed to hit the target? The solution is found numerically using Newton's method. This is a very simple implementation—it includes no convergence test, and simply performs 15 iterations of Newton's method.

After entering the program, store it into variable `PROJTL`. Then to run the program, enter: v_0 `ENTER` x_t `ENTER` y_t `ENTER` θ_0 `PROJTL`, where v_0, x_t, and y_t are in any consistent set of units, and θ_0 (the initial estimate of the launch angle) is in degrees. The program returns the launch angle θ in degrees.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
≪ → V X Y T
   ≪ RAD T D→R 'T' STO
   1 15
   START
      T T 2 * SIN X * T COS SQ Y * 2 * - X V / →NUM SQ 9.8 * -
        T 2 * COS X * 2 * T 2 * SIN Y * 2 * + / - 'T' STO
   NEXT
   T R→D DEG ≫
≫
```

Store the program into variable `PROJTL`.

Example. Let $v_0 = 30$ m/s, $(x_t, y_t) = (50\text{ m}, 20\text{ m})$, and $\theta_0 = 30°$. Enter the above program, store into variable `PROJTL`, and type `30 ENTER 50 ENTER 20 ENTER 30 PROJTL`. The program returns $\theta = 41.5357079293°$.

2 Kepler's Equation

Given the mean anomaly M (in degrees) and the orbit eccentricity e, this program solves Kepler's equation

$$M = E - e \sin E$$

to find the eccentric anomaly E. This is a very simple implementation—it includes no convergence test, and simply solves Kepler's equation by performing 15 iterations of Newton's method.

After entering the program, store it into variable `KEPLER`. Then to run the program, enter: M `ENTER` e `KEPLER`, where M is in degrees. The program returns the eccentric anomaly E in degrees.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
≪ 0 → M E EA
   ≪ RAD M D→R 'M' STO M 'EA' STO
   1 15
   START
      EA M EA - EA SIN E * + EA COS E * 1 - / - 'EA' STO
   NEXT
   EA R→D DEG ≫
≫
```

Store the program into variable `KEPLER`.

Example. Let $M = 60°$, $e = 0.15$. Enter the above program, store into variable `KEPLER`, and type `60 ENTER .15 KEPLER`. The program returns $E = 67.9667°$.

3 Hyperbolic Kepler's Equation

Given the mean anomaly M (in degrees) and the orbit eccentricity e, this program solves the hyperbolic Kepler equation

$$M = e \sinh F - F$$

to find the variable F. This is a very simple implementation—it includes no convergence test, and simply solves the hyperbolic Kepler equation by performing 15 iterations of Newton's method.

After entering the program, store it into variable `HKEPLER`. Then to run the program, enter: M `ENTER` e `HKEPLER`, where M is in degrees. The program returns the variable F.

Program Listing

```
≪ 0 → M E F
   ≪ M D→R 'M' STO M 'F' STO
   1 15
   START
      F M F + F SINH E * - 1 F COSH E * - / - 'F' STO
   NEXT
   F ≫
≫
```

Store the program into variable `HKEPLER`.

Example. Let $M = 60°$, $e = 1.15$. Enter the above program, store into variable `HKEPLER`, and type `60 ENTER 1.15 HKEPLER`. The program returns $F = 1.55551859439$.

4 Barker's Equation

Given the constant $K = \sqrt{GM/(2q^3)}(t - T_p)$, this program solves Barker's equation

$$\tan\left(\frac{f}{2}\right) + \frac{1}{3}\tan^3\left(\frac{f}{2}\right) = \sqrt{\frac{GM}{2q^3}}\,(t - T_p)$$

to find the true anomaly f.

After entering the program, store it into variable `BARKER`. Then to run the program, enter the dimensionless number K:

$$K = \sqrt{\frac{GM}{2q^3}}\,(t - T_p)$$

followed by `ENTER`. The program returns the true anomaly f in degrees.

The program will work in either degrees or radians mode.

Program Listing

```
« → K
   « K ABS 1.5 * DUP SQ 1 + √ + 3 XROOT DUP SQ 1 - SWAP /
   IF 'K<0' THEN NEG END
   ATAN 2 * »
»
```

Store the program into variable `BARKER`.

Example. Let $K = 19.38$. Enter the above program, store into variable `BARKER`, and type `19.38 BARKER`. The program returns $f = 149.084724939°$.

5 Reduction of an Angle

This program reduces a given angle to the range $[0, 360°)$ in degrees mode, or $[0, 2\pi)$ in radians mode. It will work correctly whether the calculator is set for degrees or radians mode.

After entering the program, store it into variable REDUCE. Then to run the program, enter: θ REDUCE. The program will return the equivalent reduced angle.

Program Listing

```
≪ 0 → T Z
   ≪ -1 ACOS 2 * 'Z' STO
      CASE 'T<0' THEN 'T+(IP(-T/Z)+1)*Z' END
           'T≥Z' THEN 'T-IP(T/Z)*Z' END
           T END EVAL
   ≫
≫
```

Store the program into variable REDUCE.

Example. Let $\theta = 5000°$ and set the calculator's angle mode to degrees. Enter the above program, store into variable REDUCE, and type 5000 REDUCE. The program returns 320°.

6 Helmert's Equation

Given the latitude θ (in degrees) and the elevation H (in meters), this program uses Helmert's equation to find the acceleration due to gravity g.

After entering the program, store it into variable HELMERT. Then to run the program, enter: θ ENTER H HELMERT, where θ is in degrees and H is in meters. The program returns the acceleration due to gravity g in m/s^2.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
« → θ H
   « DEG 9.80616 θ 2 * COS 0.025928 * - θ 2 * COS SQ 6.9E-5 * +
   3.086E-6 H * - »
»
```

Store the program into variable HELMERT.

Example. Let $\theta = 38.898°$, $H = 53$ m. Enter the above program, store into variable HELMERT, and type 38.898 ENTER 53 HELMERT. The program returns $g = 9.80052$ m/s^2.

7 Pendulum Period

Given the length L and amplitude θ of a simple plane pendulum, this program finds the period T, using the arithmetic-geometric mean method.

After entering the program, store it into variable PEND. Then to run the program, enter: L ENTER θ PEND, where L is in meters and θ is in degrees. The program returns the period T in seconds.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
« 1 → L,θ,T
   « DEG
   1 θ 2 / COS + 2 / 'A' STO
   θ 2 / COS √ 'G' STO
   1 10
   FOR N
      A G + 2 / 'B' STO
      A G * √ 'G' STO
      B 'A' STO
   NEXT
   L 9.8 / √ 2 * π * A / →NUM »
»
```

Store the program into variable PEND.

Example. Let $L = 1.2$ m and $\theta = 65°$. Enter the above program, store into variable PEND, and type 1.2 ENTER 65 PEND. The program returns $T = 2.3898$ sec.

8 1D Perfectly Elastic Collisions

Given the masses m_1 and m_2 of two bodies and their initial velocities v_{1i} and v_{2i}, this program finds the post-collision velocities v_{1f} and v_{2f}, using

$$v_{1f} = \left(\frac{m_1 - m_2}{m_1 + m_2}\right) v_{1i} + \left(\frac{2m_2}{m_1 + m_2}\right) v_{2i}$$
$$v_{2f} = \left(\frac{2m_1}{m_1 + m_2}\right) v_{1i} + \left(\frac{m_2 - m_1}{m_1 + m_2}\right) v_{2i}$$

After entering the program, store it into variable `ELAS1D`. Then to run the program, enter: m_1 `ENTER` m_2 `ENTER` v_{1i} `ENTER` v_{2i} `ELAS1D`. The program returns the post-collision velocities v_{1f} in the `X` register and v_{2f} in the `Y` register, in the same units.

Program Listing

```
« → M N V W
  « 2 M * V * M N + / N M - M N + / W * +
  M N - M N + / V * 2 N * W * M N + / + »
»
```

Example. Let $m_1 = 2.0$ kg, $m_2 = 7.0$ kg, $v_{1i} = 4.0$ m/s, and $v_{2i} = -5.0$ m/s. Enter the above program, store into variable `ELAS1D`, and type `2 ENTER 7 ENTER 4 ENTER 5 +/- ELAS1D`. The program returns $v_{1f} = -10$ m/s in the `X` register and $v_{2f} = -1$ m/s in the `Y` register.